
1. Постановка задачи
Необходимо написать программу, реализующую параллельную работу нескольких процессов. Каждый процесс может состоять из одного или нескольких потоков. Любой из потоков, работающих в составе этих процессов, может быть приостановлен и вновь запущен некоторой определенной клавишей (можно использовать буквенные или цифровые клавиши). Нажатия клавиш обрабатывать с помощью прерывания от клавиатуры (по материалам лаб. работы №1).

Окончание работы программы должно происходить при приостановке всех потоков их ключевыми клавишами либо при нажатии клавиши ESC. При окончании работы необходимо выполнить корректное завершение, т.е. “дочитать” всю информацию из буфера каждого процесса (при его наличии), закрыть все открытые файлы и т.п. – по материалам лаб. работы №4.

Те потоки, которые выводят информацию на экран, должны использовать для этого каждый свое отдельное окно, обрамленное рамкой – экран визуально делится на несколько окон, в каждом из которых отображается один процесс.

Дополнительное информационное окно должно содержать описание управляющих клавиш и краткое сообщение о тех действиях, которые выполняются в программе: номер потока, его статус в текущий момент времени (активен, приостановлен, ожидает), другую информацию (указание величины выделенного потоку кванта времени, процент заполненности буфера и т.п.) в зависимости от конкретного потока.

Программа должна работать устойчиво, без “зависаний” и непредвиденных зрительных эффектов. В случае необходимости указания параметров при запуске программы необходимо предусмотреть и обработать все возможные ошибки (с выводом диагностики на экран).

Задачи 1, 3, 6 предполагают наличие управляющего блока, который, используя прерывания таймера, случайным образом определяет очередной активный поток и выделяет ему кванты времени (возможно, в рамках большего кванта, выделенного всему процессу). Кванты времени – как для процесса в целом, так и для его потоков – изначально задавать некоторой фиксированной величины, но предусмотреть возможность её изменения во время работы путем нажатия некоторых ключевых клавиш (для каждого из потоков предусмотреть свою клавишу) – т.е. можно, например, увеличить, или уменьшить квант только потока-пpоизводителя. При этом допустимы разные варианты реализации – “общий” квант может либо меняться, либо оставаться постоянным (тогда при ускорении одного потока другой автоматически замедлится, т.к. величина его кванта уменьшится).

Потоки этого класса задач могут иметь три статуса: “активен”, “ожидает” или “приостановлен”. В процессе работы может возникнуть, например, следующая ситуация. Поток-потребитель приостановлен своей ключевой клавишей, следовательно, буфер не освобождается. Поток-производитель активен, он заполнит буфер информацией и перейдёт в состояние ожидания. Из этого состояния он сможет выйти только после того, как будет возобновлена (нажатием клавиши) работа потока-потребителя, который освободит место в буфере для помещения новой информации. В случае приостановки производителя возникнет аналогичная ситуация, только с пустым буфером. Приостановка/возобновление потока возможны в любом его состоянии – как в активном, так и в состоянии ожидания.

Задачи 2, 4, 5, 7 должны выполняться равномерно, независимо от степени загрузки системы. Для этого каждой из них необходимо получать управление через фиксированное количество “тиков” системного таймера, во время которого они выполнят какое-то свое элементарное действие (“бегущая строка” или “летающий объект” сместится на одну позицию, сменится нота в музыке…). При такой реализации скорость каждого потока будет определяться количеством “тиков” таймера между его запусками. Для уменьшения скорости такого потока достаточно после нажатия ключевой клавиши предоставлять ему управление реже, через большее число “тиков”, соответственно для ускорения такого потока – опять же после нажатия ключевой клавиши – ему предоставляется управление чаще, в пределе – на каждом “тике”. Так, “бегущая строка” должна двигаться равномерно с постоянной скоростью (если она не приостановлена ключевой клавишей) независимо от количества активных процессов в системе, музыка – тоже играть равномерно… Потоки этого класса задач могут иметь два статуса: “активен” и “приостановлен”.

ЗАДАЧИ:

1. Два потока: первый читает информацию из файла (например, стихи или текст программы) в буфер, второй эту информацию из буфера выдаёт на экран. При заполнении окна вывода до конца его содержимое не должно обновляться полностью – вывод новой информации должен осуществляться в последнюю строку, а все остальные строки смещаться вверх (по материалам лаб. работы №4). Имя читаемого файла задавать как параметр командной строки. После окончания файла он начинает считываться заново.

2. Музыкальное сопровождение – как минимум несколько нот различной длительности, образующие мелодию. Во включённом состоянии мелодия должна играть в фоновом режиме (по материалам лаб. работы №2).

5. Вывод заголовка работы (фамилия и имя автора, и название работы) в виде “бегущей строки” (по материалам лаб. работы №5). Место вывода (верхняя или нижняя строка экрана) задавать параметром командной строки.

7. Движущийся объект (в простейшем случае, например, летающий и отражающийся от границ окна шарик). Движение может осуществляться свободно или при управлении с клавиатуры (по материалам лаб. работы №5).

Выбор номера варианта курсовой работы осуществляется по двум последним цифрам пароля.

Описание входных данных программы и её результатов

В качестве параметров командной строки в программу передаётся два параметра:

a) положение вывода заголовка работы (бегущей строки) – (U – вверху (Up), D – внизу(Down));

b) имя файла, из которого будет считываться текст (целое число от 1 до 4). Расширение «.txt» добавляется автоматически.

При попытке запуска программы без параметров выводится сообщение об отсутствии параметров.

При некорректных параметрах выводится сообщение с подсказкой о возможных параметрах.

Пример запуска программы:

KursRab.exe u 2

Описание основных алгоритмов, используемых в программе.

Параллельная работа потоков реализована с использованием прерываний системного таймера. Потоки производитель и потребитель работают с общим буфером FIFO типа. Все операции с буфером реализованы с помощью отдельных подпрограмм. Потоку потребителю и производителю программа выделяет определённый интервал времени на работу (также есть возможность увеличить время работы потоков с помощью специальных клавиш). Поток прерывает свою работу в трёх случаях:

1) Вышло время, отведённое на работу потока.

2) Невозможно произвести операцию с буфером (для производителя это в случае, когда буфер заполнен, для потребителя – когда буфер пуст).

3) Поток остановлен нажатием специальной клавиши.

После операций добавления / извлечения элемента в / из буфер(а) производится обновление показания уровня заполненности буфера, отображаемого на экране.

Звуковое сопровождение, бегущая строка и движущийся объект в виде летающей решетки реализованы как фоновые процессы. Для каждого такого процесса есть интервал времени (задан константами MusicTR, ObjTR и выражением, вычисляемым с использованием переменной LineSpeed для соответствующих процессов), через который эти процессы производят какое-либо предусмотренное действие (сменяется частота звука, сдвигается строка, сдвигается «летающий объект»). Изменить состояние этих процессов можно с помощью специальных клавиш.

Вся информация обо всех процессах и потоках, а также клавиши управления отображаются на экране.

Описание основных переменных, констант и типов

Для описания состояния процесса в определённый момент времени реализован тип

Status = (Active, Waits, Suspended).

Константы.

LText – текст бегущей строки;

MaxQueueSize – максимальный размер очереди, используемой как буфер;

MusicTR – время смены частоты музыки;

ObjTR – время сдвига движущегося объекта;

Pause – задержка для потребителя и производителя.

Переменные.

Music, Line, Obj, Prod, Cons – состояния соответствующих потоков;

TMusic, TLine, TObj, T – счётчики времени соответствующих потоков;

TProd, TCons – дополнительное время для производителя и потребителя соответственно;

Q – очередь;

Head, Tail – голова и хвост очереди соответственно;

OldTHandler, OldKHandler – старые обработчики прерываний;

Sx, Sy – координаты движущегося объекта;

Dx, Dy – "направляющие" движущегося объекта;

Tmp_Symbol, Tmp_Attr – часть экрана, перекрытая движущимся объектом;

Len – длина бегущей строки;

Tmp – часть экрана, перекрытая бегущей строкой;

X, Y – координаты свободного места окна вывода потока-потребителя;

CPos – текущая позиция бегущей строки;

SN – номер строки экрана, в которой движется бегущая строка;

Scr – массив для прямого доступа к видеопамяти;

Init, End – флаги начала и завершения работы программы;

Thread – номер текущего активного потока;

FileName – имя считываемого файла;

Описание основных блоков (подпрограмм) программы;

Print – процедура вывода информации на экран через массив Scr;

THandler – процедура обработки прерываний системного таймера;

KHandler – процедура обработки прерываний клавиатуры;

Producer, Consumer – потоки производитель и потребитель;

Enqueue – процедура, добавляющая элемент в очередь;

Dequeue – функция, извлекающая элемент из очереди;

Empty – функция, проверяющая очередь на пустоту;

Full – функция, проверяющая очередь на полноту;

Count – функция, возвращающая число элементов очереди;

CreateWindows – процедура, заполняющая экран окнами и начальными данными;

Initialize – процедура, инициализирующая основные переменные перед запуском основных процедур;

SetCursorSize – процедура, изменяющая размер курсора.

Текст программы

Program KursRab;
{$M $1000, 0, 0}
{$R+ $S+ Q$+}
Uses
Crt, Dos;
Type
Status = (Active, Waits, Suspended);
Const
LText: String[79] = 'Курсовая работа по дисциплине "Операционные системы".Вариант 2.';
MaxQueueSize      = 15;
{ Ёмкость буфера }
MusicTR           = 5;
ObjTR             = 2;
Pause             = 5000;
{ Время задержки в милисекундах }
TProd: Word       = 30; { Время работы производителя }
TCons: Word       = 30; { Время работы потребителя   }
Var
TMusic, TLine, TObj           : Word;
Head, Tail, LineSpeed, T      : Word;
Music, Line, Obj, Prod, Cons  : Status;
Thread, SN, X, Y, Len, Sx, Sy : Byte;
OldKHandler, OldTHandler      : Procedure;
Dx, Dy, CPos                  : Shortint;
Init, Close                   : Boolean;
Q                             : Array[1..MaxQueueSize] of String[34];
Tmp                           : String[79];
Tmp_Attr                      : Array[0..1, 0..1] of Byte;
Tmp_Symbol                    : Array[0..1, 0..1] of Char;
Scr                           : Array[1..25, 1..80] Of Record
Symbol: Char;
Attr: Byte;
End Absolute $B800:$0000;
FileName




  : text;
{ Вывод строки на экран }
Procedure Print(Const S: String; Const X, Y, Color: Byte);
Var
i: Integer;
Begin
For i := X To X + Length(S) - 1 Do Begin
Scr[Y, i].Attr := Color;
Scr[Y, i].Symbol := S[i - X + 1];
End;
End;
{ Добавление элемента в очередь }
Procedure Enqueue(Const X: String);
Begin
Q[Tail] := X;
If Tail = MaxQueueSize Then Tail := 1 Else Inc(Tail);
End;
{ Извлечение элемента из очереди }
Function Dequeue: String;
Begin
Dequeue := Q[Head];
If Head = MaxQueueSize Then Head := 1 Else Inc(Head);
End;
{ Проверка очереди на пустоту }
Function Empty: Boolean;
Begin
Empty := Head = Tail;
End;
{ Проверка очереди на заполненность }
Function Full: Boolean;
Begin
If Tail < Head Then
Full := Tail + 1 = Head
Else
Full := (Head = 1) And (Tail = MaxQueueSize);
End;
{ Возвращает число элементов очереди }
Function Count: Word;
Begin
If Tail < Head Then
Count := MaxQueueSize - Head + Tail
Else
Count := Tail - Head;
End;
{ Процедура обработки прерываний системного таймера }
{$F+}
Procedure THandler; Interrupt;
Var
i, j: Integer;
Begin
Inc(T);
If Music = Active Then Begin
Inc(TMusic);
If TMusic = MusicTR Then Begin
TMusic := 0;
NoSound;
Sound(Random(6000));
End;
End;
If Line = Active Then Begin
Inc(TLine);
If TLine > 18.2 / LineSpeed Then Begin
Dec(CPos);
If Not Init Then Begin
Scr[SN, CPos + Len].Attr := White;
Scr[SN, CPos + Len].Symbol := Tmp[Length(Tmp)];
End;
Delete(Tmp, Length(Tmp), 1);
If (CPos >= 1) And (CPos + Len <= 80) Then
Tmp := Scr[SN, CPos].Symbol + Tmp
Else
Tmp := Scr[SN, 80 + CPos].Symbol + Tmp;
If (CPos >= 1) And (CPos + Len <= 80) Then
For i := CPos To CPos + Len - 1 Do Begin
Scr[SN, i].Attr := LightGreen;
Scr[SN, i].Symbol := LText[i - CPos + 1];
End;
If CPos < 1 Then Begin
If Not Init Then
For i := 1 To Len + CPos - 1 Do Begin
Scr[SN, i].Attr := LightGreen;
Scr[SN, i].Symbol := LText[-CPos + i + 1];
End;
For i := 80 + CPos To 80 Do Begin
Scr[SN, i].Attr := LightGreen;
Scr[SN, i].Symbol := LText[i - (80 + CPos) + 1];
End;
If CPos - 1 = -Len Then Begin
CPos := 80 - Len + 1;
If Init Then Init := False;
End;
End;
TLine := 0;
End;
End;
If Obj = Active Then Begin
Inc(TObj);
If TObj >= ObjTR Then Begin
While (Sx + Dx > 77) Or (Sx + Dx < 43) Or (Sy + Dy > 22) Or (Sy + Dy < 16) Do Begin
Case Random(4) Of
0: Dx := -1;
1: Dx := 1;
2: Dx := 2;
3: Dx := -2;
End;
Case Random(4) Of
0: Dy := -1;
1: Dy := 1;
2: Dy := 2;
3: Dy := -2;
End;
End;
For i := 0 To 1 Do
For j := 0 To 1 Do Begin
Scr[Sy + j, Sx + i].Attr := Tmp_Attr[j, i];
Scr[Sy + j, Sx + i].Symbol := Tmp_Symbol[j, i];
End;
Sx := Sx + Dx;
Sy := Sy + Dy;
For i := 0 To 1 Do
For j := 0 To 1 Do Begin
Tmp_Attr[j, i] := Scr[Sy + j, Sx + i].Attr;
Tmp_Symbol[j, i] := Scr[Sy + j, Sx + i].Symbol;
End;
For i := 0 To 1 Do
For j := 0 To 1 Do Begin
Scr[Sy + j, Sx + i].Attr := Green;
Scr[Sy + j, Sx + i].Symbol := Chr(35);
End;
TObj := 0;
End;
End;
Inline($9C);
OldTHandler;
End;
{$F-}
{ Процедура обработки прерываний клавиатуры }
{$F+}
Procedure KHandler; Interrupt;
Var
S: String[5];
Begin
Case Port[$60] Of
1:  Begin
Prod := Suspended;
Print('Остановлен ', 26, 10, LightRed);
Close := True;
End;
59: If Music = Active Then Begin
Music := Suspended;
NoSound;
Print('Остановлен ', 26, 5, LightRed);
End
Else Begin
Music := Active;
Print('Активен      ', 26, 5, LightGreen);
End;
60: If Line = Active Then Begin
Line := Suspended;
Print('Остановлен ', 26, 6, LightRed);
End
Else Begin
Line := Active;
Print('Активен      ', 26, 6, LightGreen);
End;
61: If Obj = Active Then Begin
Obj := Suspended;
Print('Остановлен ', 26, 7, LightRed);
End
Else Begin
Obj := Active;
Print('Активен      ', 26, 7, LightGreen);
End;
62: If Prod <> Suspended Then Begin
Prod := Suspended;
Print('Остановлен ', 26, 10, LightRed);
End
Else Begin
Prod := Active;
If Thread = 0 Then Print('Активен      ', 26, 10, LightGreen)
Else Print('Ожидание     ', 26, 10, Yellow);
End;
63: If Cons <> Suspended Then Begin
Cons := Suspended;
Print('Остановлен ', 26, 11, LightRed)
End
Else Begin
Cons := Active;
If Thread = 1 Then Print('Активен      ', 26, 11, LightGreen)
Else Print('Ожидание     ', 26, 11, Yellow);
End;
64: If TProd > 1   Then Dec(TProd);
65: If TProd < 500 Then Inc(TProd);
66: If TCons > 1   Then Dec(TCons);
67: If TCons < 500 Then Inc(TCons);
68: If LineSpeed > 1  Then Dec(LineSpeed);
87: If LineSpeed < 18 Then Inc(LineSpeed);
End;
Print('        ', 26, 16, White);
Print('        ', 26, 17, White);
Str(TProd, S);
Print(S, 26, 16, White);
Str(TCons, S);
Print(S, 26, 17, White);
Str(LineSpeed, S);
Print(S + ' ', 28, 20, White);
Inline($9C);
OldKHandler;
End;
{$F-}
{ Производитель }
Procedure Producer;
Var
Story: String[34];
S: String[8];
Begin
If (Prod = Waits) And Not Full Then Prod := Active;
Case Prod Of
Active:    Print('Активен    ', 26, 10, LightGreen);
Waits:     Print('Ожидание   ', 26, 10, Yellow);
Suspended: Print('Остановлен ', 26, 10, LightRed);
End;
T := 0;
While (T <= TProd) And (Prod <> Suspended) Do Begin
If Full Then Begin
Prod := Waits;
Print('Ожидание   ' , 26, 10, Yellow);
Break;
End;
If EOF(FileName) Then
Reset(FileName);
Read(FileName, Story);
{ чтение строки из файла }
While Length(Story) < 34 Do
Story:= Story + ' ';
Enqueue(Story);
{ добавление элемента в очередь }
Str(Count / (MaxQueueSize - 1) * 100 :3:2, S);
{ заполненность буфера }
Print(S + ' %      ', 21, 13, White);
{ вывод информации о заполненности буфера }
Delay(Pause);
End;
If Prod = Active Then Print('Время вышло', 26, 10, LightGray);
End;
{ Потребитель }
Procedure Consumer;
Var
i, j: Integer;
S: String[8];
Story: String[34];
Begin
If (Cons = Waits) And Not Empty Then Cons := Active;
Case Cons of
Active:    Print('Активен    ', 26, 11, LightGreen);
Waits:     Print('Ожидание   ', 26, 11, Yellow);
Suspended: Print('Остановлен ', 26, 11, LightRed);
End;
T := 0;
While (T <= TCons) And (Cons <> Suspended) Do Begin
If Empty Then Begin
Cons := Waits;
Print('Ожидание   ' , 26, 11, Yellow);
Break;
End;
Story:= Dequeue;
{ Извлечение элемента из очереди }
X := 44;
If Y = 12 Then Begin
For i := 3 To 11 Do
For j := 43 To 78 Do
Scr[i, j].Symbol := Scr[i + 1, j].Symbol;
For j := 43 To 78 Do
Scr[12, j].Symbol := ' ';
End
Else Inc(Y);
Print(Story, X, Y, Green);
{ вывод строки из буфера на экран }
Str(Count / (MaxQueueSize - 1) * 100 :3:2, S);
{ заполненность буфера }
Print(S + ' %      ', 21, 13, White);
{ вывод информации о заполненности буфера }
Delay(Pause);
End;
If Cons = Active Then Print('Время вышло', 26, 11, LightGray);
End;
{ Изменение размера курсора }
Procedure SetCursorSize(Const BegLine, EndLine: Byte);
Var
Regs: Registers;
Begin
With Regs Do Begin
AH := $01;
CH := BegLine;
CL := EndLine;
End;
Intr($10, Regs);
End;
{ Заполнение экрана окнами и начальными данными }
Procedure CreateWindows;
Begin
TextBackground(Black);
ClrScr;
Window(2, 2, 40, 24);
TextBackground(White);
ClrScr;
TextColor(Black);
GotoXY(11, 1);
Write('Управление процессами');
Window(3, 3, 39, 23);
TextBackground(Black);
ClrScr;
Window(3, 3, 39, 4);
TextBackground(Brown);
ClrScr;
GotoXY(9, 1);
TextColor(White);
WriteLn('Управление состояниями');
GotoXY(11, 2);
Write('фоновых  процессов');
Window(3, 5, 39, 7);
TextBackground(Black);
ClrScr;
TextColor(White);
WriteLn('F1 Музыка            :');
WriteLn('F2 Бегущая строка    :');
Write('F3 Движущийся объект :');
Window(3, 8, 39, 9);
TextBackground(Brown);
ClrScr;
TextColor(White);
GotoXY(9, 1);
WriteLn('Управление состояниями');
GotoXY(10, 2);
TextColor(White);
Write('параллельных потоков');
Window(3, 10, 39, 11);
TextBackground(Black);
ClrScr;
TextColor(White);
WriteLn('F4 Производитель     :');
Write('F5 Потребитель       :');
Window(3, 12, 39, 12);
TextBackground(Brown);
ClrScr;
TextColor(White);
GotoXY(12, 1);
Write('Состояние буфера');
Window(3, 13, 39, 13);
TextBackground(Black);
ClrScr;
Write('Буфер заполнен на');
Window(3, 14, 39, 15);
TextBackground(Brown);
ClrScr;
GotoXY(4, 1);
WriteLn('Время потоков ["тиков" таймера]');
GotoXY(7, 2);
Write('<уменьшить> / <увеличить>');
Window(3, 16, 39, 17);
TextBackground(Black);
ClrScr;
TextColor(White);
WriteLn('F6 / F7 Производитель:');
Write('F8 / F9 Потребитель  :');
Window(3, 18, 39, 18);
TextBackground(Brown);
ClrScr;
TextColor(White);
GotoXY(7, 1);
Write('Другие клавиши управления');
Window(3, 19, 39, 23);
TextBackground(Black);
ClrScr;
TextColor(White);
WriteLn('F10 / F11 - скорость бегущей строки');
WriteLn('Текущая скорость строки:');
WriteLn;
Write('Esc - выход');
Window(42, 2, 79, 13);
TextBackground(White);
ClrScr;
TextColor(Black);
GotoXY(9, 1);
Write('Вывод потока-потребителя');
Window(43, 3, 78, 12);
TextBackground(Black);
ClrScr;
Window(42, 15, 79, 24);
TextBackground(White);
ClrScr;
TextColor(Black);
GotoXY(9, 1);
Write('Окно летающего объекта');
Window(43, 16, 78, 23);
TextBackground(Black);
ClrScr;
End;
{ Инициализация основных переменных }
Procedure Initialize;
Var
i, j: Integer;
S: String;
Begin
SetCursorSize($20, $00);
Init := True;
Close := False;
Music := Active;
Obj := Active;
Line := Active;
Prod := Active;
Cons := Waits;
TMusic := 0;
TLine := 0;
TObj := 0;
CPos := 1;
Len := Length(LText);
LineSpeed := 5;
X := 43;
Y := 3;
Tail := 1;
Head := 1;
Print('Активен', 26, 5, LightGreen);
Print('Активен', 26, 6, LightGreen);
Print('Активен', 26, 7, LightGreen);
Print('Не запущен', 26, 10, LightGray);
Print('Не запущен', 26, 11, LightGray);
Print('0 %', 21, 13, White);
Str(TProd, S);
Print(S, 26, 16, White);
Str(TCons, S);
Print(S, 26, 17, White);
Str(LineSpeed, S);
Print(S, 28, 20, White);
Tmp := '';
For i := 1 To Len Do
Tmp := Tmp + Scr[SN, i].Symbol;
Dx := -1;
Dy := -1;
Sx := 60;
Sy := 20;
For i := 0 To 1 Do
For j := 0 To 1 Do Begin
Tmp_Attr[j, i] := Scr[SY + j, SX + i].Attr;
Tmp_Symbol[j, i] := Scr[SY + j, SX + i].Symbol;
End;
End;
Var
i, j: Integer;
S, F: String[1];
Begin
S := ParamStr(1);
F := ParamStr(2);
If (ParamCount = 0) Then Begin
WriteLn('Не введены параметры командной строки.');
Halt;
End;
If (Not (Upcase(S[1]) In ['D', 'U'])) Or (Not (F[1] In ['1','2','3','4'])) Then Begin
ClrScr;
WriteLn('Неправильно введены параметры командной строки.');
WriteLn('Должно быть два параметра.');
WriteLn('Первый параметр - положение бегущей строки:');
WriteLn(' U - вверху, D - внизу');
WriteLn('Второй параметр - имя считываемого файла:');
WriteLn('целое число от 1 до 4');
Halt;
End;
If Upcase(S[1]) = 'U' Then SN := 1 Else SN := 25;
Assign(Filename, F + '.txt');
{ Устанавливается связь файловой переменной с файлом }
Reset(Filename);
{ Открытие файла для чтения }
CreateWindows;
Initialize;
Randomize;
GetIntVec($9, @OldKHandler);
SetIntVec($9, Addr(KHandler));
GetIntVec($8, @OldTHandler);
SetIntVec($8, Addr(THandler));
While Not Close Do Begin
Thread := Random(2);
Case Thread Of
0: Producer;
1: Consumer;
End;
End;
Prod := Suspended;
Cons := Active;
While Not Empty Do
Consumer;
NoSound;
SetCursorSize($06, $07);
Window(1, 1, 80, 25);
GotoXY(1, 25);
SetIntVec($9, Addr(OldKHandler));
SetIntVec($8, Addr(OldTHandler));
End.

Результаты тестирования

